Prologprogramm `primrek.pl` zur Berechnung des Wertes eines primitivrekursiven Funktionals ϕ an einer Stelle $(x_1, \ldots, x_n)$

Der Zweck dieses Programmes ist der Nachweis, dass die primitivrekursiven Funktionen in reinem Hornklausel-Prolog ohne Verwendung eingebauter Prädikate und somit in einem Kalkül der reinen Prädikatenlogik erster Stufe berechenbar sind.

1 Datenstrukturen

1.1 Natürliche Zahlen

Struktur	Darstellung in Prolog
Natürliche Zahl n	$\underbrace{\texttt{f(...f}}_{n \text{ mal}}\texttt{(0)...)}$

1.2 primitivrekursive Funktionale

1. `O` ist ein 0-stelliges Funktional. `O` heißt das *Nullfunktional*.	`o`
2. `N` ist ein 1-stelliges Funktional. `N` heißt das *Nachfolgerfunktional*.	`n`
3. $\mathtt{P}^n_i$ ist ein n-stelliges Funktional für $i, n \in \mathbb{N}$ mit $1 \leq i \leq n$. $\mathtt{P}^n_i$ heißt das i-te n-stellige *Projektionsfunktional*.	$\mathtt{p(}i\mathtt{)}$
4. $(\phi\psi_1 \ldots \psi_m)$ ist ein n-stelliges Funktional, wenn ϕ ein m-stelliges Funktional ist und $\psi_1, \ldots, \psi_m$ n-stellige Funktionale sind. $(\phi\psi_1 \ldots \psi_m)$ heißt die *Komposition* von ϕ mit $\psi_1, \ldots, \psi_m$.	$\mathtt{[}\phi\mathtt{,}\psi_1\mathtt{,...,}\psi_n\mathtt{]}$
5. $(\mathtt{R}\phi\psi)$ ist ein $(n+1)$-stelliges Funktional, wenn ϕ ein n-stelliges und ψ ein $(n+2)$-stelliges Funktional ist. $(\mathtt{R}\phi\psi)$ heißt das durch *primitive Rekursion* aus ϕ und ψ gebildete Funktional.	$\mathtt{r(}\phi\mathtt{,}\psi\mathtt{)}$

Bemerkung: Wird der Wert eines Funktionals, das als Teilfunktional ein Projektionsfunktional $\mathtt{P}_i^n$ enthält, an einer Stelle berechnet, so kann die Stelligkeit n von $\mathtt{P}_i^n$ indirekt aus der Anzahl der Argumente von $\mathtt{P}_i^n$ erschlossen werden und braucht daher in der Implementierung nicht explizit repräsentiert zu werden.

2 Das Prädikat `wert/3`

`wert(`ϕ`,[`x_1`,...,`x_n`],`W`)` bedeutet, dass W der Wert des n-stelligen Funktionals ϕ an der Stelle $(x_1, \ldots, x_n)$ ist.

3 Beispiel für eine Anfrage

Wir wollen den Wert des Additionsfunktionals add $= (\mathtt{RP}_1^1(\mathtt{NP}_1^3))$ an der Stelle $(2, 3)$ berechnen lassen. Hierzu müssen wir zunächst das Additionsfunktional als Prologterm darstellen wie oben beschrieben: `r(p(f(0)),[n,p(f(0))])` und ebenso die Zahlen 2 und 3: `f(f(0))` bzw. `f(f(f(0)))`. Dann stellen wir die Anfrage

```
?- wert(r(p(f(0)),[n,p(f(0))]),[f(f(0)),f(f(f(0)))],W).
```

Darauf liefert Prolog die Antwort

```
W = f(f(f(f(f(0)))))
```

`f(f(f(f(f(0)))))` ist die Darstellung der Zahl 5 als Term der reinen Prädikatenlogik erster Stufe.

4 Prologprädikate zur Listenverarbeitung

Die Konkatenation der Listen L_1 und L_2 ist die Liste L	`konk(`L_1`,`L_2`,`L`)`
Das n-te Element der Liste L ist x	`tes(`n`,`L`,`x`)`

4.1 Prologklauseln für diese Prädikate

Konkatenation zweier Listen

```
konk([],Ys,Ys).
konk([X|Xs],Ys,[X|Zs]):-
   konk(Xs,Ys,Zs).
```

Auswahl des n-ten Elementes einer Liste

```
tes(f(0),[X|Xs],X).
tes(f(N),[Y|Xs],X):-
   tes(N,Xs,X).
```

5 Der Wert eines Funktionals an einer Stelle

Das Funktional ϕ hat bei $(x_1, \ldots, x_n)$ den Wert w.	`wert(`ϕ`,[`x_1`,...,`x_n`],`w`)`
Die Funktionale $\phi_1, \ldots, \phi_m$ haben bei $(x_1, \ldots, x_n)$ die Werte $w_1, \ldots, w_m$.	`werte([`ϕ_1`,...,`ϕ_m`],[`x_1`,...,`x_n`],[`w_1`,...,`w_m`])`

5.1 Induktive Definition des Wertes eines Funktionals an einer Stelle, und Prologklauseln dafür

1. `O` hat bei () den Wert 0.

```
wert(o,[],0).
```

2. `N` hat bei x den Wert x'.

```
wert(n,[X],f(X)).
```

3. $\mathtt{P}_i^n$ hat bei $(x_1, \ldots, x_n)$ den Wert x_i.

```
wert(p(I),Xs,X):-
   tes(I,Xs,X).
```

4. $(\phi\psi_1 \ldots \psi_m)$ hat bei $(x_1, \ldots, x_n)$ den Wert z, wenn ψ_i bei $(x_1, \ldots, x_n)$ den Wert y_i hat für $i = 1, \ldots, m$ und ϕ bei $(y_1, \ldots, y_m)$ den Wert z hat.

```
wert([Phi|Psis],Xs,Z):-
   werte(Psis,Xs,Ys),
   wert(Phi,Ys,Z).
```

5. $(\mathtt{R}\phi\psi)$ hat bei $(x_1, \ldots, x_n, 0)$ den Wert z, wenn ϕ bei $(x_1, \ldots, x_n)$ den Wert z hat.

```
wert(r(Phi,Psi),Vs,Z):-
   konk(Xs,[0],Vs),
   wert(Phi,Xs,Z).
```

6. $(\mathtt{R}\phi\psi)$ hat bei $(x_1, \ldots, x_n, y')$ den Wert z, wenn $(\mathtt{R}\phi\psi)$ bei $(x_1, \ldots, x_n, y)$ den Wert u hat und ψ bei $(u, x_1, \ldots, x_n, y)$ den Wert z hat.

```
wert(r(Phi,Psi),Vs,Z):-
   konk(Xs,[f(Y)],Vs),
   konk(Xs,[Y],Us),
   wert(r(Phi,Psi),Us,U),
   wert(Psi,[U|Us],Z).
```

(s. oben)

```
werte([],Xs,[]).
werte([Phi|Phis],Xs,[W|Ws]):-
   wert(Phi,Xs,W),
   werte(Phis,Xs,Ws).
```